
Specific exceptions to the above are hereby granted to all individuals and entities:

Visual Basic 32-bit Application Development


Standards and Guidelines





1.	INTRODUCTION


These development standards and guidelines cover a wide range of topics.  They are intended to provide a framework for consistent development and maintainable code.  Specific applications may require additions to these guidelines for application unique requirements.  These additions should be documented and maintained during the life cycle of the application.





These standards and guidelines are intended for 32-bit applications being developed for the various Microsoft Windows operating systems using Microsoft's Visual Basic.





2.	REFERENCES


The user interface design will follow the guidelines specified in Microsoft Windows 95 User Interface Design Guide.  Developers who are involved in user interface design are strongly encouraged to read About Face, The Essentials of User Interface Design by Alan Cooper.





All developers are encouraged to read Code Complete, A Practical Handbook of Software Construction by Steve McConnell.  The sections on commenting code and code reviews are particularly important.





The Visual Basic Programmer’s Guide to the Win32 API, by Daniel Appleman, is an excellent resource.  Developers should consider this book as part of their Visual Basic documentation.





3.	PROGRAMMING CONSTRUCTS


3.1.	Naming Conventions


3.1.1.	Basic Rules


The intent of these naming conventions is to permit any developer to understand the important characteristics of any given object without having to search through code to ascertain that information.  Each class of objects uses specific conventions to provide this information.  In addition, the Visual Basic development language imposes certain naming rules.  These rules are:





Names must begin with a letter


Names must contain only letters and/or numbers


Names may contain the underscore (_) character but not spaces or other punctuation marks


Names can be as long as 40 characters


Visual Basic reserved words may not be used as names.





Each name will use a predefined prefix to identify the object or variable type.  These prefixes are defined in the sections that follow.  All prefixes will be lower case letters.  The first character following the prefix in a name will be an upper case letter.  The use of upper and lower case letters, numbers, and underscores in names is encouraged to improve readability.  While names may be up to 40 characters in length, developers are encouraged to use the minimum length possible without sacrificing readability.  Name lengths in the range of 9 to 15 characters are considered optimal.





Names should be chosen that are clear and unambiguous.  Names should reflect the real world nature of the object rather than use computer or data processing terms.  Abbreviations should be avoided.  In the case of Boolean objects, the name should reflect the object's TRUE condition.





Naming conventions for Active-X components are described in “Active-X Component Standards and Guidelines” under “Creating Active-X Components” in Microsoft’s Component Tools Guide.  This manual is provided in the Visual Basic Books Online.  These conventions are considered to be part of this document.





3.1.2.	Classes, Forms, Modules, etc.


All classes will be named according to the “Active-X Component Standards and Guidelines” under “Creating Active-X Components” in Microsoft’s Component Tools Guide.  Class, form, module, user control, and user document names (the Name property, not the file name) will use a single upper case letter prefix.  Classes will have “C” as the prefix except when the class is an interface class.  Interface classes will have “I” as the prefix.  Forms will have “F” and modules will have “M” as their respective prefixes, while the user control prefix will be “U”, and the user document prefix will be “D”.





File names will be chosen appropriate to the file.  ActiveX component file names should typically use a “ihs” prefix to easily identify IHS Professional Services’ components.  Also, when creating ActiveX components, several settings on the General tab of the Project Properties dialog need special treatment.  The Project Name should be chosen as appropriate to the project, but the name must have a “ihs” prefix.  The Project Name is displayed in the Project/Library dropdown list in the Object Browser.  The Project Description should start with “IHS” and use proper name capitalization (capitalize like a title of a book).  The Project Description is used to identify a component in the References dialog.





3.1.3.	Controls


When Visual Basic controls are created, they have default names provided automatically by the development environment.  These default names WILL NOT be used as the control name.  Instead, create a name using a prefix to identify the type of control with a name that adequately describes the function or purpose of the control.  The prefix for standard Visual Basic controls will consist of three (3) lower case letters followed by the name.  The prefix for third party custom controls will consist of three (3) lower case letters followed by an additional lower case letter or number (chosen as a unique identifier for the set of custom controls) followed by the name.  If the custom control is a replacement for a standard control, the first three letters in the prefix should be the same as the standard control.  If the custom control is a unique control, a three-letter prefix must be chosen and used consistently throughout the application.  The above requirements apply to controls created by or for IHS Professional Services, and the fourth letter of the prefix shall be “i”.





Control names should follow the basic rules described above; however, due to their unique hierarchical nature, menus have additional requirements.  Captions on the menu bar must be one word to conform to Windows standards.  Captions of menu options and sub menu options should be no longer than two words.  Menus should normally have only one level of sub menu options under any given menu option.  Menu names consist of the prefix followed by the menu bar caption and the menu option caption.  For example, the Exit menu option under the File menu would have a menu name of mnuFileExit.  For menu options with two word captions, remove the space between the two words.  For example, the Special Paste menu option under the Edit menu would have a menu name of mnuEditSpecialPaste.  To distinguish between menu options and sub menu options, insert an underscore character between their names.  For example, the Link sub menu option under the Special Paste menu option would have a menu name of mnuEditSpecialPaste_Link.





The following table contains prefixes for the Visual Basic controls.





Control�
Prefix�
Example�
�
3D Panel�
pnl�
pnlStatusBar�
�
Animated Button�
ani�
aniPhoneConnect�
�
Check Box�
chk�
chkReadOnly�
�
Combo Box�
cbo�
cboTypeCode�
�
Command Button�
cmd�
cmdOK�
�
Common Dialog�
dlg�
dlgBrowse�
�
Communications�
com�
comCommPort1�
�
Crystal Reports Control�
rpt�
rptMailingLabels�
�
Data Control�
dat�
datClient�
�
DBCombo (data-bound)�
dbc�
dbcState�
�
DBGrid (data-bound)�
dbg�
dbgRegister�
�
DBList (data-bound)�
dbl�
dblTitle�
�
Directory List Box�
dir�
dirSource�
�
Drive List Box�
drv�
drvTarget�
�
File List Box�
fil�
filSource�
�
Form�
frm�
frmAboutDialog�
�
Frame�
fra�
fraUserOptions�
�
Gauge�
gau�
gauProgress�
�
Graph�
gra�
graSalesYTD�
�
Grid�
grd�
grdToDo�
�
Horizontal Scroll Bar�
hsb�
hsbVolume�
�
Image�
img�
imgIcon�
�
ImageList�
ils�
ilsToolBarImages�
�
ImageList Item�
ili�
iliToolBarButton�
�
Key State�
key�
keyCapsLock�
�
Label�
lbl�
lblLastName�
�
Line�
lin�
linDivider�
�
List Box�
lst�
lstStates�
�
List View�
lvw�
lvwNames�
�
List View Item�
lvi�
lviUserName�
�
MAPI Message�
mpm�
mpmMAPIMessage�
�
MAPI Session�
mps�
mpsMAPISession�
�
Masked Edit�
msk�
mskWorkPhone�
�
Menu Option�
mnu�
mnuFileExit�
�
Multimedia MCI�
mci�
mciMMControl�
�
OLE Client�
ole�
oleWordDoc�
�
Option Button�
opt�
optPrefix�
�
Outline�
out�
outGroupMembers�
�
Pen BEdit�
bed�
bedName�
�
Pen HEdit�
hed�
hedSignature�
�
Pen Ink�
ink�
inkMap�
�
Picture Box�
pic�
picToolBar�
�
Picture Clip�
clp�
clpToolBar�
�
ProgressBar�
prg�
prgPrinting�
�
RichTextBox�
rtf�
rtfNotes�
�
Shape�
shp�
shpSquare�
�
Slider�
sld�
sldVolume�
�
Spin Button�
spn�
spnCopies�
�
StatusBar�
sta�
staMessage�
�
Tab�
tab�
tabSettings�
�
TabStrip�
tbs�
tbsOptions�
�
Text Box�
txt�
txtLastName�
�
Timer�
tmr�
tmrTrigger�
�
ToolBar�
tlb�
tlbMain�
�
TreeView�
tvw�
tvwPartList�
�
TreeView Node object�
tvn�
tvnPart�
�
UpDown�
upd�
updVolume�
�
Vertical Scroll Bar�
vsb�
vsbMouseSensitiviy�
�



3.1.4.	Variables


The variable name for all variables, including arrays, will consist of a two (2) lower case letter prefix followed by a name.  The first letter of the prefix will indicate the scope of the variable, and the second letter will indicate the variable data type.  Local variables will NOT use a scope indicator.  Parameters in public class property or method procedures will NOT use either the scope or data type prefixes.  





The letters that make up the prefixes are:





	First Letter (scope)


	Public (global)		g


	Form			f


	Class			c


	Module		m


	Local			(none)


	Parameter		p	(arguments in procedures and functions)





	Second Letter (data type)


	Boolean		b


	Byte			w


	Integer			i


	Long			l


	Single			s


	Double			d


	Currency		c


	String (text)		t


	Date(Time)		x


	Variant			v





	Special Case(objects)


	Form			frm	(parameter passed "As Form")


	Control		ctl	(parameter passed "As Control";


					Use the control prefix for specific


					controls such as ‘txt’ for “As Text”)


	Object			obj


	Collection		col	(collection of objects)


	User Control		axc


	User Document	axd


	ActiveX Data Objects (ADO)


	Connection		cnn


	Command		cmd


RecordSet		rst


Field			fld


Parameter		prm


Property		pro


Error			err





	Examples:


	Form level integer variable		fiObjectCount


	Module lever single variable		msPercentComplete


	Form level currency variable		fcInvoiceTotal


	Class level long variable		clItemCounter


	Local integer variable			iCounter


	Parameter Boolean variable		pfShowItem


	Parameter control			pctlSourceControl


	Parameter textbox			ptxtSourceTextBox





3.1.5.	Enumerations


Enumerations will be declared with the enumeration type in all capital letters with their components following normal variable conventions except that the enumeration prefix will be used in place of the scope and datatype prefix.  The enumeration prefix will be a two to five letter prefix for identifying the enumeration, and it will be placed in an in-line comment on the ‘Enum’ line. It also will be used for the data type portion of a variable prefix.





	Example:


	Public Enum RETURNCODE			‘rc


	    rcSuccessWithInfo = 1


	    rcSuccess = 0


	    rcFailure = -1


	    rcNoDataFound = -7


	    rcNoMoreData = -8


	End Enum





	Private mrcReturn As RETURNCODE





3.1.6.	User Defined Types (Structures)


User defined types will be declared with the data type in all capital letters with their components following normal variable conventions except that no scope prefix will be used.  A two to five letter prefix for identifying the user defined type will be placed in an in-line comment on the ‘Type’ line.  It will be used for the data type portion of the variable prefix.





	Example:


	Type EMPLOYEE				‘emp


	    tName As String


	    tAddress As String


	    tCityStateZip As String


	    xHireDate As Date


	    cSalary As Currency


	End Type





	Private mempEmployee As EMPLOYEE





3.1.7.	Special Context Tags


When a variable is used in a special context, the tag should reflect that context even if it is one of the above variable types.  Typical context tags are shown below.





	Context Tags:


	Handle			h		Integer


	Window Handle	hwnd		Integer


	Device Handle		hdc		Integer


	Return Code		rc		Integer





	Examples:


	Handle			hCursor


	Window Handle	hwndForm


	Device Handle		hdcPrinter


	Return Code		rcStatus





Other context tags may be used but must be documented in the <EXENAME>.BAS module.





It is permissible to use just the context tag without a name (i.e., ‘rc’ rather than ‘rcStatus’) when the use of the context tag is completely unambiguous.  If there is any chance for confusion, the full tag plus name convention will be used.





3.1.8.	Constants


All non-variable data used in the application must be defined as named constants.  Intrinsic VB and VBA constants will be used wherever applicable.  These constants can be found in the VB help file.  Of particular interest are constants for special characters.  Use vbCrLf [Chr$(13) + Chr$(10)], vbCr [Chr$(13)], vbLf [Chr$(10)], vbBack [Chr$(8)], vbTab [Chr$(9)], and vbNullChar [Chr$(0)] rather than the equivalent Chr$ functions.  Other predefined constants, such as API and type constants, should be used whenever possible.  





Application specific constants will be named as appropriate using all uppercase letters, numbers, and underscore (_) characters with a single lowercase letter prefix for data type.  Context tags may be used in place of data type tags.  If a group of constants is related, they should have a common prefix of up to five (5) upper case letters and/or numbers followed by an underscore (_) character.  The common prefix will follow the data type prefix.





	Examples:


	String constant	tAPP_NAME


	Long constant		lCLR_BLACK


	Return Code		rcFAILURE





3.1.9.	General Procedures


General procedures consist of user defined Sub procedures and Function procedures.  General procedures can be either public or private depending on where the procedure is placed or how it is declared.  Class property procedures are public by definition.  Class method procedures are just public Sub or Function procedures.





To distinguish between public and private general procedures, a single letter prefix will be added to all private general procedures.  For form general procedures, the prefix will be a lower case "f"; for private module procedures, the prefix will be “m”; for private class procedures, the prefix will be “c”.  Public procedures will not use a prefix.  In addition, function procedures will use a lower case prefix to indicate the data type that the function returns.  The function data type prefix will use the same letters as the variable data type prefix, or context tag, specified above.  The data type prefix will follow the scope prefix.





Class property procedures and public Sub and Function procedures used as class methods will NOT use the scope or datatype prefixes.  In addition, parameters of public property or public method procedures will NOT use the scope or datatype prefixes.  Names will follow standard Windows conventions for properties and methods.  Refer to “Active-X Component Standards and Guidelines” under “Creating Active-X Components” in Microsoft’s Component Tools Guide for clarification.  All method names will follow the “verb/object” naming convention.





NOTE:  All event procedures are generated by Visual Basic for forms and controls.  These event procedures are private to the form containing them and are automatically named based on the form or control name and the event.  The above naming conventions do not apply to event procedures.  However, the above conventions do apply to events that are created as part of classes by a developer.





3.2.	Variable Declarations


All form, class, and module level variables, including arrays, will be explicitly declared as a specific data type in the declarations section of the appropriate form, class, or module.  These variables should be declared with the Private keyword rather than the Dim keyword.  Public variables should rarely, if ever, be used!





All variables local to a Sub procedure or Function procedure must be declared explicitly in the beginning of the procedure using the Dim or Static keywords as appropriate.





Variables should normally be declared as one of the defined data types rather than the variant data type.  Unless a particular requirement dictates the use of a variant data type, variant data type variables should be avoided.  





Parameters (procedure arguments) should be explicitly declared as specific data types in the parameter list of the procedure declaration.  Be sure to use the ByVal keyword as appropriate.





3.3.	Named Constants


Named constants should be used to the maximum extent possible.  The use of named constants will greatly improve the readability and maintainability of the source code.  More importantly, if future modifications are necessary, the constants can be changed in one location.





The exception to using named constants is to use Enumerations wherever a group of related constants that represent long integer datatypes are required.





3.4.	Path Names


Path names will not be hard coded into any application.  Provisions must be made that permit the application to ascertain or permit the user to select any required file path information.  All path names should use the Universal Naming Convention (UNC) (\\server\share\directory) for shared drives.  Mapped drive letters should be avoided unless provided as input from the user.





3.5.	Comments/Documentation


Each form, module, and class will have a standard header comment section as the first item in the declarations section.  The header will contain the file name of the form or module, the copyright notice, the creation date, the original developer, and a clear description of the form, module, or class.





Each procedure will have a standard header comment section immediately following the Sub, Function, or Property statement.  The header for general procedures will contain a clear description of the procedure and a description of each parameter and any return values.  A maintenance line will also be included containing the date of the change activity, the developer’s name, and a brief description of the change activity.  During development, only a line for procedure creation will be used.  When the application enters the maintenance phase (after ‘code complete’), the maintenance lines will be included.  These headers will be placed in all user defined general procedures.





The header for event procedures will only contain maintenance lines containing the date of the change activity, the developer’s name, and a brief description of the change activity.  These headers will be placed in all event procedures that contain code.  During development, only a line for procedure creation will be used.  When the application enters the maintenance phase (after ‘code complete’), the maintenance lines will be included.  They are NOT to be placed in event procedures that DO NOT contain code.


Code, when well written, should be self-documenting; however, self-documented code cannot possibly describe the developer’s intent or explain an algorithm or section of logic!  Therefore, comments are required to ensure that the code can be maintained by communicating this type of information.  Comments must communicate information and not what code is coming next.  Developers should document their code liberally with in-line comments.  Comments should allow a different developer to understand the purpose and function of the code including its relationship to other code modules.  





In-line comments should precede the source code that it references and be indented to the same level as the source code.  While discouraged, very short end-line comments may be appended to the end of a source code line if the line can be viewed in the standard VB code window.  The end-line comment must be separated from the code by at least one full tab stop.





Refer to the section on commenting in Code Complete, A Practical Handbook of Software Construction by Steve McConnell.  This section gives practical guidance on how to comment code.





3.6.	Code Formatting


3.6.1.	Visual Basic Options/Environment


The options found under the Tools/Options menu may be set as the developer chooses except for:


	In the Editor tab:


		Auto Syntax Check			Checked


		Require Variable Declaration		Checked


		Auto Indent				Checked


		Tab Width				4





3.6.2.	Indentation


Code in all procedures will be indented one (1) tab stop.  Code inside programming constructs such as If...EndIf, Select Case...End Select, For...Next, and Do...Loop will be indented one (1) additional tab stop.  Nested constructs will be indented an additional tab stop for each level of nesting.





3.6.3.	White Space


Use white space and individual or double blank lines to indicate logical program sections.  Blank lines should separate programming constructs such as If...EndIf and For...Next from surrounding code.  Blank lines should also be used to improve readability of the code such as in large Select Case...End Select constructs.





3.6.4.	Comments


File Headers


File headers will be placed as the first entry in the declarations section of the file.  Option Explicit must immediately follow the header.  The declaration sections follow the Option Explicit statement. Declarations will be in the order shown in the sample file header.  All public declarations should proceed file level declarations in each section where such distinctions can be made.  See Appendix A for the formatting of all file headers.





Procedure Headers


Procedure headers will follow the procedure declaration.  A variable declaration section will immediately follow the procedure header.  Constants will come first in the declarations section followed by variables.  See Appendix A for the formatting of all procedure headers.





In-Line Comments


In-line comments for describing a particular process or a particular segment of code will consist of the comment block with a formatted comment line before and after it.  Blank lines should come before and after the comment block.  The formatted comment line should contain approximately 45 dashes (-) to segregate the comment from the surrounding code.  Use this style of comment block for all comments that do not describe particular lines of code.  It will be constructed as follows:


'---------------------------------------------


'Comment goes here.  Use as many lines as


'necessary, being sure to start a new line so


'the comment is visible in a typically sized


'code window.  One formatted comment line should


'precede and follow the actual comment.


'---------------------------------------------





In-line comments for describing particular lines of code, or a group of code lines such as loops, will consist of the comment block with no special comment line before or after it.  A blank comment line should precede the comment block, and there should be no blank lines between the comment block and the code to which it refers.  It will be constructed as follows:


'Comment goes here.  Use as many lines as necessary,


'being sure to start a new line so the comment is


'visible in a typically sized code window





Code Maintenance Comments


Code maintenance comments will not be used until the ‘code complete’ milestone.  After that point, the following commenting will be utilized.





A code maintenance section will be appended to the file header under the Code Maintenance Log section to identify and summarize the changes made to the file.  Use the following format:





'


'Developer:		Joe Programmer


'Change Date:	06/05/98


'Reference:		Put in, as appropriate, any problem


'			report or bug references.  If the


'			change is an enhancement, indicate it


'			as such.


'Description:	Put in a brief description of the


'			change(s) made.


'---------------------------------------------





If the version control tools being used provide the capability to automatically add this information, use the tool’s format rather than the one above.





In the procedure where the changes are made, the developer must add an entry to the history section of the procedure header to include the developer’s name, the date of the change, and a brief comment describing the changes including any bug numbers.  Refer to the example below:





'-----------------------------------------------------------------


'Date        Developer           Comments


'03/15/98    J. Programmer       Initial creation


'05/24/98    J. Programmer       Bug 1234; Changed column names to


'                                new naming standards


'


'*****************************************************************





At the actual location in the code where the changes are made, the developer is to add a comment block showing “**”, their initials, the date, and the bug number or other reference in the first comment line.  Use subsequent comment lines to describe the actual changes. .  Refer to the example below:





'** JP 05/24/98 – Bug 1234


'** Change the query columns in the SQL statement


'** to correspond to the new naming standards.





NOTE: Rather than deleting code, the developer should comment out the code for future reference.  The developer must exercise their judgement on the necessity of re-writing the procedure when commented out, or “deleted”, code causes the procedure to no longer be maintainable.  In that case, the developer should re-write the procedure and add an entry to the history section documenting the re-write.  DO NOT remove other history entries.





4.	STANDARD CONVENTIONS


4.1.	Global Modules


While the majority of modules in an application will be class modules, there will be certain standard modules, described below, for each application.  The number of global modules will be based on the needs of the application, but should be minimized in favor of class modules.  NOTE:  These modules do not apply to ActiveX components.





4.1.1.	<EXENAME>.BAS


This module, named as the application EXE, will contain all documentation for special naming conventions including application specific contexts.  In addition, it will contain any application specific public constants.





4.1.2.	BEGINEND.BAS


This module will contain the Sub Main() and Sub ExitApp() procedures along with related startup and shutdown procedures.  All applications will have their startup in Sub Main().  All applications will have a single exit point in Sub ExitApp() containing all necessary housekeeping code required for a normal application shutdown.  All application exit points must call this procedure.  





4.1.3.	INC_<VEND>.BAS


This “include” module will contain all general-purpose third party vendor API procedure and function declarations and any related global constants and type definitions.  There will be one module for each vendor or vendor’s product as appropriate. This module will be a shared in the version control tool among the various projects that use the particular third party tool.





4.2.	Coding Conventions


Developers are responsible for designing and writing code that takes size, speed, and maintainability into consideration.  These three concerns are of equal priority.  To ensure these three concerns are balanced, developers must thoroughly understand their code.  They should step through the code and/or use a profiling tool to see the actual process flow and the procedures where the larger amounts of time are spent.  Developers should try multiple approaches when performance is an issue.  Timing the different approaches, running the different approaches on a low end computer, and using other third party tools, such as a profiler, can all be used to determine the best approach.  Once the approach is chosen, the developer can optimize the code for speed and size.  Regardless of the speed or size of the resulting code, the code must be maintainable!





Coding conventions are provided to assist the developer in meeting the requirement to balance the above considerations.  They have been developed through experience and testing.  The coding conventions found in Appendix B have been adopted to balance application performance, application size, and maintainability for applications developed with the current version of Visual Basic.  This appendix may be updated upon the release of subsequent versions of Visual Basic.





Developers should also acquaint themselves with the white paper entitled Optimizing Microsoft Visual Basic 4.0.  It is equally applicable to subsequent versions of Visual Basic.  It can be found on the MSDN Library CD.





4.3.	Message Boxes


Message boxes will be used throughout applications to provide feedback to the user and to solicit a response from the user for specific purposes.  Three types of message boxes, described below, will be used.  A fourth type, the query (question) message box, should not be used.  It currently exists in Visual Basic for backward compatibility.  All message boxes will be invoked in code with a message, the appropriate message box parameters, and the full application name.  The message box parameters, using standard Visual Basic constants, will be concatenated to obtain the required results. The MsgBox statement will be used whenever the user just needs to acknowledge the message by clicking on the OK button.  When the user is presented with multiple options such as with Yes, No, Cancel buttons, the MsgBox function will be used in order to trap for the return value corresponding to the button the user selected.





4.3.1.	Information Message Box





�


Information messages should be used to report the results of an action or to provide feedback to the user that cannot otherwise be provided.  An example of what an information message box should look like, and the code that invoked it, is shown below:�





tMsg = "No records were found to satisfy the query."


MsgBox tMsg, vbOKOnly + vbInformation, tAppName





4.3.2.	Warning Message Box


Warning messages should be used to inform the user that a non-critical action cannot be performed or an abnormal result or condition occurred.  An example of what a warning message box should look like, and the code that invoked it, is shown below:





�








tMsg = "The required file could not be located."


MsgBox tMsg, vbOKOnly + vbExclamation, tAppName





4.3.3.	Critical Message Box


Critical messages should be used to inform the user that a critical action cannot be performed or an abnormal result or condition occurred that may cause the application to abort.  An example of what a critical message box should look like, and the code that invoked it, is shown below:








�





tMsg = "The database connection was lost during processing."


MsgBox tMsg, vbOKOnly + vbCritical, tAppName





4.4.	Error Handling


Virtually all procedures will utilize an error trap for expected and unexpected run time errors.  Developers may exercise their judgment and decide that a general or event procedure is simple enough that error handling is not necessary; however, all procedures in ActiveX components must utilize error traps.  Handling of those errors is left to the developer’s judgement.





Every effort will be made to handle errors within a procedure.  When this is not possible, error conditions will be raised back up to the calling procedure utilizing appropriate implementations of the Err.Raise method available in a standard IHS Professional Services ActiveX component.  Refer to the Programmer’s Guide in the VB documentation for a discussion on error handling with ActiveX objects.





4.5.	Single Exit Point


All procedures, regardless of their type will have one and only one exit point.  This is one of the very few times that GoTo may be used.  If a procedure has multiple conditions for an exit, each condition should end with a GoTo statement.  The GoTo label for this case should be the procedure name with 'Exit' appended to the end.  The exit label should be positioned just before the section of housekeeping code prior to the exit.





4.6.	Class Initialization


The Visual Basic form and class modules have a Class object having two events, Initialize and Terminate.  These events are fired only once each at the instantiation and termination of the object represented by the class.  Unfortunately, the Initialize event does not have any parameters, therefore, it can not be used to initialize the object using passed-in data that is available only at run time.  To get around this limitation, any class having the requirement to be initialized using data not available to the object will have a method named InitializeObject.  This method may have any parameters necessary to complete initialization of the object.  It is to be invoked immediately following the creation of the object.








�
APPENDIX A: HEADER FORMATS


File Header





'Workfile:		ANYFORM.FRM


'Copyright:		Copyright © 1998 IHS Professional Services


'Created:		03/15/98


'Author:		Joe Programmer


'Description:	Description of file; Include sufficient


'                 info to clearly describe the component


'Dependencies:    List any other dependent components


'Issues:          List known issues, problems, etc.


'


Option Explicit





'=============================================


'Code Maintenance Log 


'=============================================








'=============================================


'Type Definition Declaration Section


'=============================================








'=============================================


'Enumeration Declaration Section


'=============================================








'=============================================


'API/DLL Procedure Declaration Section


'=============================================








'=============================================


'Constant Declaration Section


'=============================================








'=============================================


'Variable Declaration Section


'=============================================


























Const EndCommentBlock = 0





Event Procedure Header





Use only for event procedures having code.





'-----------------------------------------------------------------


'Date        Developer           Comments


'03/15/98    J. Programmer       Initial creation


'


'*****************************************************************


    '=============================================


    'Local Constant/Variable Declaration Section


    '=============================================


    


    





User Defined Procedure Header





Use for all property, method, general function, and general sub procedures.





'Description:   


'Pass:          N/A


'Return:        N/A


'Raise Errors:  N/A


'-----------------------------------------------------------------


'Date        Developer           Comments


'03/15/98    J. Programmer       Initial creation


'


'*****************************************************************


    '=============================================


    'Local Constant/Variable Declaration Section


    '=============================================


    


    


File Header Code Maintenance Log





Place this comment block under the Code Maintenance Log section in the file header for each maintenance action.  Use AFTER the “code complete” milestone.





'


'Developer:	     Joe Programmer


'Change Date:    06/05/98


'Reference:	     Put in, as appropriate, any problem


'                report or bug number references.  


'Description:    Put in a brief description of the


'                change(s) made.


'---------------------------------------------





�
APPENDIX B:  CODING CONVENTIONS FOR VISUAL BASIC


Use of Option Explicit





Option Explicit must be used in every form and module.  It must be the first code line following the form/module header but before the declarations sections.  Use of Option Explicit will require the developer to explicitly declare all variables thereby eliminating errors introduced by using misspelled variables.  It also forces the developer to declare a variable as a specific data type.





Use of “&” and “+” for Concatenation





Whenever concatenating strings use the concatenation operator (“&”) rather than the plus sign (“+”).  This will prevent unexpected results should variants be used.





Use of An Object’s Default Property





When referring to an object without referencing a property, the default property of the object is being invoked.  For example:





	lblFormLabel.Caption = “Something”





and





	lblFormLabel = “Something”





are equivalent.  The first line uses a specific property to set the caption while the second line is using the label’s default property to set the caption.  Using an object's default property makes the code less maintainable.  Developers must remember each object’s default property or must look it up in the documentation to be sure they understand what property is being set.  Therefore, developers are to always use an object’s specific property rather than its default property.





Use of GoTo





The use of GoTo can make code more difficult to follow and more difficult to maintain.  Therefore, GoTo should only be used for the specific purposes below.





GoTo should be used as a clause in the On Error statement.  When used, the On Error statement should be the first line of code following the variable declarations in a procedure.  The GoTo label should be made up of the procedure name with 'VBErr' appended to the end.  





When an error condition is detected within the code (not from a runtime error), a different GoTo label made up of the procedure name with 'Err' appended to the end may be used if the error handling requirements are different between detected error conditions and runtime errors.





GoTo should also be used to enforce a single exit point from a procedure.  If a procedure has multiple conditions for an exit, each condition should end with a GoTo statement.  The GoTo label for this case should be the procedure name with 'Exit' appended to the end.  The exit label should be positioned just before the section of housekeeping code prior to the exit.





Comparisons





Conditional tests must normally be based on comparisons of string, integer (Boolean and enumerated), and long integer data types.  Equality comparison of any floating point data type is not recommended.  When making comparisons to named constants, be sure the variable or expression will result in the same data type as the named constant.





When comparing strings, always trim leading and trailing blank characters and convert the string to upper case. For example:





	If UCase$(Trim$(ftVariable)) = "VALUE" Then





When checking a Boolean integer data type, be aware that Visual Basic uses '-1' for exactly True and '0' for exactly False.  Using an expression that yields other than these values may not behave as expected, especially when using logical operators.  When checking a condition based on a Boolean expression or value, do not compare it to True or False.  For example, use:





	If (fiValue >= 1 And fiValue <= 10) Then





rather than





	If (fiValue >= 1 And fiValue <= 10) = True Then





If the Boolean expression is long or complex, it should be set to a Boolean variable and the Boolean variable used in the condition.  For example,





	fInRange = (fiValue >= 1 And fiValue <= 10)


	If fInRange Then





If...EndIf Constructs





There are four basic If...EndIf constructs.  The first is known as an Immediate If (IIf).  It is a Visual Basic function available in VB version 3.0.  Use this function when making a simple assignment based on the results of a Boolean expression.  Unless the expression is simple and short, it is recommended that the results of the expression be assigned to a Boolean variable and use the Boolean variable in the IIf function.  For example:





	fInRange = (fiRange >= 0 And fiRange <=100)


	tCaption = IIf(fInRange,"Passed","Failed")





The second construct is the If...EndIf.  This construct is used when a specific action is required based on a single specific condition.  While VB allows a single line construct, these guidelines require a multi-line construct.  For example, use:





	If ffLoadingForm Then


	    GoTo Form_ResizeExit


	EndIf





rather than





	If ffLoadingForm Then GoTo Form_ResizeExit





This greatly enhances maintainability by making the code more readable and improves debugging when single stepping through the code.





The third construct is the If...Else...EndIf.  This construct is used when specific actions are required based on whether a single specific condition is True or False.  For example:





	If (fiValue >= 1 And fiValue <= 10) Then


	    lblFormLabel.Caption = “In Range”


	Else


	    lblFormLabel.Caption = “Out of Range”


	EndIf





The fourth construct is the If   ElseIf...ElseIf...EndIf.  This construct is used when a specific action is required based on multiple specific and often unrelated conditions.  It is important that the conditions are properly ordered with the highest priority action listed first.  For example:





	If ffLoadingForm Then


	    GoTo ProcNameExit





	ElseIf chkDoIt.Value = vbChecked


	    Call DoItProc





	ElseIf Trim$(txtField.Text) = “” Then


	    Call WarnEmptyProc





	Else


	    Call NothingWorkedProc





	EndIf





Select Case Constructs





This construct is used when various specific actions are required based on the value of a specific expression having multiple possible values.  For example:





	Select Case fiIndex


	Case 0


	    Call fDoCase0





	Case 1


	    Call fDoCase1





	Case 2


	    Call fDoCase2





	Case Else


	    Call fDoAnythingElse





	End Select





The above format is the preferred format.  The alternate format below is also acceptable.





	Select Case fiIndex


	    Case 0


	        Call fDoCase0





	    Case 1


	        Call fDoCase1





	    Case 2


	        Call fDoCase2





	    Case Else


	        Call fDoAnythingElse





	End Select








�
APPENDIX C:  EXAMPLE PROCEDURE


Public Function ConvertMessage(ByVal MessageString As String, _�                               ByVal ParameterList As String, _�                               ByVal Delimiter As String) As String


'Description:   Convert a message string with parameter placeholders into


'               a string containing the parameters;  All parameter


'               place holders will be "|1", "|2", etc and the linefeed


'               placeholder will be "|/"


'Pass:          MessageString       Message string with placeholders


'               ParameterList       Message parameters in delimited string


'               Delimiter           Delimiter string used in ParameterList


'Return:        ConvertMessage      Converted message


'Raise Errors:  N/A


'-----------------------------------------------------------------


'Date       Developer       Comments


'05/19/98   J.Paquette      Initial Creation


'


'-****************************************************************


    '=============================================


    'Local Constant/Variable Declaration Section


    '=============================================


    Const tPLACEHOLDER_LF = "|/"             'Linefeed place holder


    


    Dim tParameterList As String


    Dim tDelimiter As String


    Dim vParamArray As Variant


    Dim tMsg As String


    Dim iParam As Integer


    Dim tPlaceHolder As String


    Dim tParameter As String





    On Error GoTo ConvertMessageVBErr


    


    'Default to the source string


    ConvertMessage = MessageString





    'Validate parameters


    tParameterList = Trim$(ParameterList)


    tDelimiter = Trim$(Delimiter)


    


    If Len(tParameterList) = 0 Or Len(tDelimiter) = 0 Then


        GoTo ConvertMessageExit


    End If


    


    'Parse the parameter list


    vParamArray = ParseStringArray(tParameterList, tDelimiter, vbTextCompare)


    If IsNull(vParamArray) Then


        GoTo ConvertMessageExit


    End If





    'Build the message string


    tMsg = MessageString





    'Replace all parameter placeholders with their parameters


    For iParam = 1 To UBound(vParamArray)


        tPlaceHolder = "|" & Format$(iParam)


        tParameter = CStr(vParamArray(iParam))


        tMsg = ReplaceSubString(tMsg, tPlaceHolder, tParameter, vbTextCompare)


    Next 'iParam





    'Replace all linefeed place holders with linefeeds


    tMsg = ReplaceSubString(tMsg, tPLACEHOLDER_LF, vbLf, vbTextCompare)





    ConvertMessage = tMsg





ConvertMessageExit:


    Exit Function





ConvertMessageVBErr:


    'Any error, return MessageString


    ConvertMessage = MessageString


    Resume ConvertMessageExit





End Function

